

Defining (Co)datatypes in Isabelle/HOL

Jasmin Christian Blanchette, Martin Desharnais,
Lorenz Panny, Andrei Popescu, and Dmitriy Traytel
Fakultät für Informatik, Technische Universität München

27 August 2014

Abstract

This tutorial describes the new package for defining datatypes and codatatypes in Isabelle/HOL. The package provides four main commands: `datatype_new`, `codatatype`, `primrec`, and `primcorec`. The first command is expected to eventually replace the old `datatype` command.

Contents

1	Introduction	3
2	Defining Datatypes	5
2.1	Introductory Examples	5
2.1.1	Nonrecursive Types	5
2.1.2	Simple Recursion	5
2.1.3	Mutual Recursion	6
2.1.4	Nested Recursion	6
2.1.5	Auxiliary Constants and Properties	7
2.2	Command Syntax	8
2.2.1	<code>datatype_new</code>	8
2.2.2	<code>datatype_compat</code>	11
2.3	Generated Constants	12
2.4	Generated Theorems	13
2.4.1	Free Constructor Theorems	13

2.4.2	Functorial Theorems	15
2.4.3	Inductive Theorems	17
2.5	Compatibility Issues	18
3	Defining Recursive Functions	19
3.1	Introductory Examples	19
3.1.1	Nonrecursive Types	19
3.1.2	Simple Recursion	20
3.1.3	Mutual Recursion	21
3.1.4	Nested Recursion	22
3.1.5	Nested-as-Mutual Recursion	23
3.2	Command Syntax	23
3.2.1	primrec	23
3.3	Recursive Default Values for Selectors	24
3.4	Compatibility Issues	25
4	Defining Codatatypes	26
4.1	Introductory Examples	26
4.1.1	Simple Corecursion	26
4.1.2	Mutual Corecursion	27
4.1.3	Nested Corecursion	27
4.2	Command Syntax	27
4.2.1	codatatype	27
4.3	Generated Constants	28
4.4	Generated Theorems	28
4.4.1	Coinductive Theorems	29
5	Defining Corecursive Functions	30
5.1	Introductory Examples	31
5.1.1	Simple Corecursion	31
5.1.2	Mutual Corecursion	32
5.1.3	Nested Corecursion	33
5.1.4	Nested-as-Mutual Corecursion	34
5.1.5	Constructor View	35
5.1.6	Destructor View	36
5.2	Command Syntax	37
5.2.1	primcorec and primcorecursive	37
6	Introducing Bounded Natural Functors	38
6.1	Bounded Natural Functors	39
6.2	Introductory Examples	39

1	<i>Introduction</i>	3
6.3	Command Syntax	41
6.3.1	bnf	41
6.3.2	bnf_axiomatization	42
6.3.3	print_bnfs	43
7	Deriving Destructors and Theorems for Free Constructors	44
7.1	Command Syntax	44
7.1.1	free_constructors	44

1 Introduction

The 2013 edition of Isabelle introduced a new definitional package for freely generated datatypes and codatatypes. The datatype support is similar to that provided by the earlier package due to Berghofer and Wenzel [1], documented in the Isar reference manual [9]; indeed, replacing the keyword **datatype** by **datatype_new** is usually all that is needed to port existing theories to use the new package.

Perhaps the main advantage of the new package is that it supports recursion through a large class of non-datatypes, such as finite sets:

```
datatype_new 'a treefs = Nodefs (lblfs: 'a) (subfs: "'a treefs fset")
```

Another strong point is the support for local definitions:

```
context linorder
begin
datatype_new flag = Less | Eq | Greater
end
```

Furthermore, the package provides a lot of convenience, including automatically generated discriminators, selectors, and relators as well as a wealth of properties about them.

In addition to inductive datatypes, the new package supports coinductive datatypes, or *codatatypes*, which allow infinite values. For example, the following command introduces the type of lazy lists, which comprises both finite and infinite values:

```
codatatype 'a llist = LNil | LCons 'a "'a llist"
```

Mixed inductive–coinductive recursion is possible via nesting. Compare the following four Rose tree examples:

```
datatype_new 'a treeff = Nodeff 'a "'a treeff list"
datatype_new 'a treefi = Nodefi 'a "'a treefi llist"
codatatype 'a treeif = Nodeif 'a "'a treeif list"
```

codatatype 'a tree_{ii} = Node_{ii} 'a "'a tree_{ii} llist"

The first two tree types allow only paths of finite length, whereas the last two allow infinite paths. Orthogonally, the nodes in the first and third types have finitely many direct subtrees, whereas those of the second and fourth may have infinite branching.

The package is part of *Main*. Additional functionality is provided by the theory *BNF_Axiomatization*, located in the directory `~/src/HOL/Library`.

The package, like its predecessor, fully adheres to the LCF philosophy [4]: The characteristic theorems associated with the specified (co)datatypes are derived rather than introduced axiomatically.¹ The package is described in a number of papers [2, 3, 7, 8]. The central notion is that of a *bounded natural functor* (BNF)—a well-behaved type constructor for which nested (co)recursion is supported.

This tutorial is organized as follows:

- Section 2, “Defining Datatypes,” describes how to specify datatypes using the **datatype_new** command.
- Section 3, “Defining Recursive Functions,” describes how to specify recursive functions using **primrec**.
- Section 4, “Defining Codatatypes,” describes how to specify codatatypes using the **codatatype** command.
- Section 5, “Defining Corecursive Functions,” describes how to specify corecursive functions using the **primcorec** and **primcorecursive** commands.
- Section 6, “Introducing Bounded Natural Functors,” explains how to use the **bnf** command to register arbitrary type constructors as BNFs.
- Section 7, “Deriving Destructors and Theorems for Free Constructors,” explains how to use the command **free_constructors** to derive destructor constants and theorems for freely generated types, as performed internally by **datatype_new** and **codatatype**.

The command **datatype_new** is expected to replace **datatype** in a future release. Authors of new theories are encouraged to use the new commands, and maintainers of older theories may want to consider upgrading.

Comments and bug reports concerning either the tool or this tutorial should be directed to the authors at `blanchette@in.tum.de`, `desharna@in.tum.de`, `lorenz.panny@in.tum.de`, `popescua@in.tum.de`, and `traytel@in.tum.de`.

¹However, some of the internal constructions and most of the internal proof obligations are skipped if the *quick_and_dirty* option is enabled.

2 Defining Datatypes

Datatypes can be specified using the `datatype_new` command.

2.1 Introductory Examples

Datatypes are illustrated through concrete examples featuring different flavors of recursion. More examples can be found in the directory `~/src/HOL/BNF/Examples`.

2.1.1 Nonrecursive Types

Datatypes are introduced by specifying the desired names and argument types for their constructors. *Enumeration* types are the simplest form of datatype. All their constructors are nullary:

```
datatype_new trool = Truee | Faalse | Perhaaps
```

Here, *Truee*, *Faalse*, and *Perhaaps* have the type *trool*.

Polymorphic types are possible, such as the following option type, modeled after its homologue from the *Option* theory:

```
datatype_new 'a option = None | Some 'a
```

The constructors are *None* :: 'a *option* and *Some* :: 'a ⇒ 'a *option*.

The next example has three type parameters:

```
datatype_new ('a, 'b, 'c) triple = Triple 'a 'b 'c
```

The constructor is *Triple* :: 'a ⇒ 'b ⇒ 'c ⇒ ('a, 'b, 'c) *triple*. Unlike in Standard ML, curried constructors are supported. The uncurried variant is also possible:

```
datatype_new ('a, 'b, 'c) triple_u = Triple_u "'a * 'b * 'c"
```

Occurrences of nonatomic types on the right-hand side of the equal sign must be enclosed in double quotes, as is customary in Isabelle.

2.1.2 Simple Recursion

Natural numbers are the simplest example of a recursive type:

```
datatype_new nat = Zero | Suc nat
```

Lists were shown in the introduction. Terminated lists are a variant that stores a value of type 'b at the very end:

```
datatype_new ('a, 'b) tlist = TNil 'b | TCons 'a "'a, 'b) tlist"
```

2.1.3 Mutual Recursion

Mutually recursive types are introduced simultaneously and may refer to each other. The example below introduces a pair of types for even and odd natural numbers:

```
datatype_new even_nat = Even_Zero | Even_Suc odd_nat
and odd_nat = Odd_Suc even_nat
```

Arithmetic expressions are defined via terms, terms via factors, and factors via expressions:

```
datatype_new ('a, 'b) exp =
  Term "('a, 'b) trm" | Sum "('a, 'b) trm" "('a, 'b) exp"
and ('a, 'b) trm =
  Factor "('a, 'b) fct" | Prod "('a, 'b) fct" "('a, 'b) trm"
and ('a, 'b) fct =
  Const 'a | Var 'b | Expr "('a, 'b) exp"
```

2.1.4 Nested Recursion

Nested recursion occurs when recursive occurrences of a type appear under a type constructor. The introduction showed some examples of trees with nesting through lists. A more complex example, that reuses our *option* type, follows:

```
datatype_new 'a btree =
  BNode 'a "'a btree option" "'a btree option"
```

Not all nestings are admissible. For example, this command will fail:

```
datatype_new 'a wrong = W1 | W2 "'a wrong  $\Rightarrow$  'a"
```

The issue is that the function arrow \Rightarrow allows recursion only through its right-hand side. This issue is inherited by polymorphic datatypes defined in terms of \Rightarrow :

```
datatype_new ('a, 'b) fun_copy = Fun "'a  $\Rightarrow$  'b"
datatype_new 'a also_wrong = W1 | W2 "('a also_wrong, 'a) fun_copy"
```

The following definition of *'a*-branching trees is legal:

```
datatype_new 'a ftree = FTLeaf 'a | FTNode "'a  $\Rightarrow$  'a ftree"
```

And so is the definition of hereditarily finite sets:

```
datatype_new hfset = HFSet "hfset fset"
```

In general, type constructors (a_1, \dots, a_m) *t* allow recursion on a subset of their type arguments a_1, \dots, a_m . These type arguments are called *live*; the

remaining type arguments are called *dead*. In $'a \Rightarrow 'b$ and $('a, 'b) \text{ fun_copy}$, the type variable $'a$ is dead and $'b$ is live.

Type constructors must be registered as BNFs to have live arguments. This is done automatically for datatypes and codatatypes introduced by the **datatype_new** and **codatatype** commands. Section 6 explains how to register arbitrary type constructors as BNFs.

Here is another example that fails:

```
datatype_new 'a pow_list = PNil 'a | PCons "( 'a * 'a ) pow_list"
```

This attempted definition features a different flavor of nesting, where the recursive call in the type specification occurs around (rather than inside) another type constructor.

2.1.5 Auxiliary Constants and Properties

The **datatype_new** command introduces various constants in addition to the constructors. With each datatype are associated set functions, a map function, a relator, discriminators, and selectors, all of which can be given custom names. In the example below, the familiar names *null*, *hd*, *tl*, *set*, *map*, and *list_all2*, override the default names *is_Nil*, *un_Cons1*, *un_Cons2*, *set_list*, *map_list*, and *rel_list*:

```
datatype_new (set: 'a) list =
  null: Nil
| Cons (hd: 'a) (tl: "'a list")
for
  map: map
  rel: list_all2
where
  "tl Nil = Nil"
```

The types of the constants that appear in the specification are listed below.

```
Constructors:  Nil :: 'a list
               Cons :: 'a  $\Rightarrow$  'a list  $\Rightarrow$  'a list
Discriminator: null :: 'a list  $\Rightarrow$  bool
Selectors:     hd  :: 'a list  $\Rightarrow$  'a
               tl  :: 'a list  $\Rightarrow$  'a list
Set function:  set :: 'a list  $\Rightarrow$  'a set
Map function:  map :: ('a  $\Rightarrow$  'b)  $\Rightarrow$  'a list  $\Rightarrow$  'b list
Relator:       list_all2 :: ('a  $\Rightarrow$  'b  $\Rightarrow$  bool)  $\Rightarrow$  'a list  $\Rightarrow$  'b list  $\Rightarrow$  bool
```

The discriminator *null* and the selectors *hd* and *tl* are characterized by

the following conditional equations:

$$\text{null } xs \implies xs = \text{Nil} \quad \neg \text{null } xs \implies \text{Cons } (\text{hd } xs) (\text{tl } xs) = xs$$

For two-constructor datatypes, a single discriminator constant is sufficient. The discriminator associated with *Cons* is simply $\lambda xs. \neg \text{null } xs$.

The **where** clause at the end of the command specifies a default value for selectors applied to constructors on which they are not a priori specified. Here, it is used to ensure that the tail of the empty list is itself (instead of being left unspecified).

Because *Nil* is nullary, it is also possible to use $\lambda xs. xs = \text{Nil}$ as a discriminator. This is the default behavior if we omit the identifier *null* and the associated colon. Some users argue against this, because the mixture of constructors and selectors in the characteristic theorems can lead Isabelle’s automation to switch between the constructor and the destructor view in surprising ways.

The usual mixfix syntax annotations are available for both types and constructors. For example:

```
datatype_new ('a, 'b) prod (infixr "*" 20) = Pair 'a 'b

datatype_new (set: 'a) list =
  null: Nil ("[]")
| Cons (hd: 'a) (tl: "'a list") (infixr "#" 65)
for
  map: map
  rel: list_all2
```

Incidentally, this is how the traditional syntax can be set up:


```
syntax "_ list" :: "args  $\Rightarrow$  'a list" ("[_]")

translations
  "[x, xs]" == "x # [xs]"
  "[x]" == "x # []"
```


2.2 Command Syntax

2.2.1 datatype_new


datatype_new : *local_theory* \rightarrow *local_theory*



dt-options



map-rel



The `datatype_new` command introduces a set of mutually recursive datatypes specified by their constructors.

The syntactic entity `target` can be used to specify a local context (e.g., *in linorder* [9]), and `prop` denotes a HOL proposition.


The optional target is optionally followed by one or both of the following options:

- The *discs_sels* option indicates that discriminators and selectors should be generated. The option is implicitly enabled if names are specified for discriminators or selectors.
- The *no_code* option indicates that the datatype should not be registered for code generation.


The optional **where** clause specifies default values for selectors. Each proposition must be an equation of the form $un_D (C \dots) = \dots$, where C is a constructor and un_D is a selector.

The left-hand sides of the datatype equations specify the name of the type to define, its type parameters, and additional information:

dt-name



tyargs




The syntactic entity *name* denotes an identifier, *mixfix* denotes the usual parenthesized mixfix notation, and *typefree* denotes fixed type variable (*'a*, *'b*, ...) [9].

The optional names preceding the type variables allow to override the default names of the set functions ($set1_t, \dots, setM_t$). Type arguments can be marked as dead by entering “*dead*” in front of the type variable (e.g., “(*dead* *'a*)”); otherwise, they are live or dead (and a set function is generated or not) depending on where they occur in the right-hand sides of the definition.

Declaring a type argument as dead can speed up the type definition but will prevent any later (co)recursion through that type argument.


Inside a mutually recursive specification, all defined datatypes must mention exactly the same type variables in the same order.

dt-ctor



The main constituents of a constructor specification are the name of the constructor and the list of its argument types. An optional discriminator name can be supplied at the front to override the default, which is $\lambda x. x = C_j$ for nullary constructors and $t.is_C_j$ otherwise.

dt-ctor-arg




The syntactic entity *type* denotes a HOL type [9].

In addition to the type of a constructor argument, it is possible to specify a name for the corresponding selector to override the default name ($un_C_j i$). The same selector names can be reused for several constructors as long as they share the same type.

2.2.2 datatype_compat

datatype_compat : *local_theory* \rightarrow *local_theory*



The `datatype_compat` command registers new-style datatypes as old-style datatypes. For example:

```
datatype_compat even_nat odd_nat
```

```
ML {* Datatype_Data.get_info @{theory} @{type_name even_nat} *}
```

The syntactic entity *name* denotes an identifier [9].

The command is interesting because the old datatype package provides some functionality that is not yet replicated in the new package, such as the integration with Quickcheck.

A few remarks concern nested recursive datatypes:

- The old-style, nested-as-mutual induction rule and recursor theorems are generated under their usual names but with “*compat_*” prefixed (e.g., *compat_tree.induct*).
- All types through which recursion takes place must be new-style datatypes or the function type. In principle, it should be possible to support old-style datatypes as well, but this has not been implemented yet (and there is currently no way to register old-style datatypes as new-style datatypes).
- The recursor produced for types that recurse through functions has a different signature than with the old package. This might affect the behavior of some third-party extensions.

An alternative to `datatype_compat` is to use the old package’s `rep_datatype` command. The associated proof obligations must then be discharged manually.

2.3 Generated Constants

Given a datatype $(a_1, \dots, a_m) t$ with $m > 0$ live type variables and n constructors $t.C_1, \dots, t.C_n$, the following auxiliary constants are introduced:

Case combinator:	$t.case_t$ (rendered using the familiar <i>case-of</i> syntax)
Discriminators:	$t.is_C_1, \dots, t.is_C_n$
Selectors:	$t.un_C_1 1, \dots, t.un_C_1 k_1$
	⋮
	$t.un_C_n 1, \dots, t.un_C_n k_n$
Set functions:	$t.set1_t, \dots, t.setm_t$
Map function:	$t.map_t$
Relator:	$t.rel_t$
Recursor:	$t.rec_t$
Size function:	$t.size_t$

The case combinator, discriminators, and selectors are collectively called *destructors*. The prefix “*t.*” is an optional component of the names and is normally hidden.

In addition to the above, the *size* class is instantiated to overload the *size* function based on *t.size_t*.

2.4 Generated Theorems

The characteristic theorems generated by **datatype_new** are grouped in three broad categories:

- The *free constructor theorems* (Section 2.4.1) are properties of the constructors and destructors that can be derived for any freely generated type. Internally, the derivation is performed by **free_constructors**.
- The *functorial theorems* (Section 2.4.2) are properties of datatypes related to their BNF nature.
- The *inductive theorems* (Section 2.4.3) are properties of datatypes related to their inductive nature.

The full list of named theorems can be obtained as usual by entering the command **print_theorems** immediately after the datatype definition. This list normally excludes low-level theorems that reveal internal constructions. To make these accessible, add the line

```
declare [[bnf_note_all]]
```

to the top of the theory file.

2.4.1 Free Constructor Theorems

The free constructor theorems are partitioned in three subgroups. The first subgroup of properties is concerned with the constructors. They are listed below for *'a list*:

```
t.inject [iff, induct_simp]:
  (x21 # x22 = y21 # y22) = (x21 = y21 ∧ x22 = y22)

t.distinct [simp, induct_simp]:
  [] ≠ x21 # x22
  x21 # x22 ≠ []

t.exhaust [cases t, case_names C1 ... Cn]:
  [y = [] ⇒ P; ∧x21 x22. y = x21 # x22 ⇒ P] ⇒ P
```

t.nchotomy:

$$\forall list. list = [] \vee (\exists x21 x22. list = x21 \# x22)$$

In addition, these nameless theorems are registered as safe elimination rules:

t.distinct [THEN notE, elim!]:

$$\begin{aligned} [] = x21 \# x22 &\implies R \\ x21 \# x22 = [] &\implies R \end{aligned}$$

The next subgroup is concerned with the case combinator:

t.case [simp, code]:

$$\begin{aligned} (\text{case } [] \text{ of } [] \Rightarrow f1 \mid x \# xa \Rightarrow f2 \ x \ xa) &= f1 \\ (\text{case } x21 \# x22 \text{ of } [] \Rightarrow f1 \mid x \# xa \Rightarrow f2 \ x \ xa) &= f2 \ x21 \ x22 \end{aligned}$$

t.case_cong [fundef_cong]:

$$\begin{aligned} \llbracket list = list'; list' = [] \implies f1 = g1; \bigwedge x21 \ x22. list' = x21 \# x22 \implies \\ f2 \ x21 \ x22 = g2 \ x21 \ x22 \rrbracket \implies (\text{case } list \text{ of } [] \Rightarrow f1 \mid x21 \# x22 \Rightarrow \\ f2 \ x21 \ x22) &= (\text{case } list' \text{ of } [] \Rightarrow g1 \mid x21 \# x22 \Rightarrow g2 \ x21 \ x22) \end{aligned}$$

t.weak_case_cong [cong]:

$$list = list' \implies (\text{case } list \text{ of } [] \Rightarrow f1 \mid x \# xa \Rightarrow f2 \ x \ xa) = (\text{case } list' \text{ of } [] \Rightarrow f1 \mid x \# xa \Rightarrow f2 \ x \ xa)$$

t.split:

$$\begin{aligned} P (\text{case } list \text{ of } [] \Rightarrow f1 \mid x \# xa \Rightarrow f2 \ x \ xa) &= ((list = [] \longrightarrow P \ f1) \\ \wedge (\forall x21 \ x22. list = x21 \# x22 \longrightarrow P \ (f2 \ x21 \ x22))) \end{aligned}$$

t.split_asm:

$$P (\text{case } list \text{ of } [] \Rightarrow f1 \mid x \# xa \Rightarrow f2 \ x \ xa) = (\neg (list = [] \wedge \neg P \ f1) \vee (\exists x21 \ x22. list = x21 \# x22 \wedge \neg P \ (f2 \ x21 \ x22)))$$

t.splits = split split_asm

The third subgroup revolves around discriminators and selectors:

t.disc [simp]:

$$\begin{aligned} \text{null } [] \\ \neg \text{null } (x21 \# x22) \end{aligned}$$

t.discI:

$$\begin{aligned} list = [] \implies \text{null } list \\ list = x21 \# x22 \implies \neg \text{null } list \end{aligned}$$

t.sel [simp, code]:

$$\begin{aligned} \text{hd } (x21 \# x22) &= x21 \\ \text{tl } (x21 \# x22) &= x22 \end{aligned}$$

t.collapse [*simp*]:

$$\begin{aligned} \text{null list} &\Longrightarrow \text{list} = [] \\ \neg \text{null list} &\Longrightarrow \text{hd list} \# \text{tl list} = \text{list} \end{aligned}$$

t.disc_exclude [*dest*]:

These properties are missing for 'a list because there is only one proper discriminator. Had the datatype been introduced with a second discriminator called *nonnull*, they would have read thusly:

$$\begin{aligned} \text{null list} &\Longrightarrow \neg \text{nonnull list} \\ \text{nonnull list} &\Longrightarrow \neg \text{null list} \end{aligned}$$

t.disc_exhaust [*case_names* $C_1 \dots C_n$]:

$$\llbracket \text{null list} \Longrightarrow P; \neg \text{null list} \Longrightarrow P \rrbracket \Longrightarrow P$$

t.sel_exhaust [*case_names* $C_1 \dots C_n$]:

$$\llbracket \text{list} = [] \Longrightarrow P; \text{list} = \text{hd list} \# \text{tl list} \Longrightarrow P \rrbracket \Longrightarrow P$$

t.expand:

$$\begin{aligned} \llbracket \text{null list} = \text{null list}'; \llbracket \neg \text{null list}; \neg \text{null list} \rrbracket \Longrightarrow \text{hd list} = \text{hd list}' \\ \wedge \text{tl list} = \text{tl list}' \rrbracket \Longrightarrow \text{list} = \text{list}' \end{aligned}$$

t.sel_split:

$$\begin{aligned} P (\text{case list of } [] \Rightarrow f1 \mid x \# xa \Rightarrow f2 \ x \ xa) = ((\text{list} = [] \longrightarrow P \ f1) \\ \wedge (\text{list} = \text{hd list} \# \text{tl list} \longrightarrow P \ (f2 \ (\text{hd list}) \ (\text{tl list})))) \end{aligned}$$

t.sel_split_asm:

$$\begin{aligned} P (\text{case list of } [] \Rightarrow f1 \mid x \# xa \Rightarrow f2 \ x \ xa) = (\neg (\text{list} = [] \wedge \neg P \\ f1 \vee \text{list} = \text{hd list} \# \text{tl list} \wedge \neg P \ (f2 \ (\text{hd list}) \ (\text{tl list})))) \end{aligned}$$

t.case_eq_if:

$$(\text{case list of } [] \Rightarrow f1 \mid x \# xa \Rightarrow f2 \ x \ xa) = (\text{if null list then } f1 \ \text{else} \\ f2 \ (\text{hd list}) \ (\text{tl list}))$$

In addition, equational versions of *t.disc* are registered with the [*code*] attribute.

2.4.2 Functorial Theorems

The functorial theorems are partitioned in two subgroups. The first subgroup consists of properties involving the constructors or the destructors and either a set function, the map function, or the relator:

t.set [*simp*, *code*]:

$$\begin{aligned} \text{set } [] &= \{\} \\ \text{set } (x21a \# x22) &= \{x21a\} \cup \text{set } x22 \end{aligned}$$

t.set_empty:

$$\overline{\text{null list}} \Longrightarrow \text{set list} = \{\}$$

t.sel_set:

$$\neg \overline{\text{null a}} \Longrightarrow \text{hd a} \in \text{set a}$$

$$\llbracket \neg \overline{\text{null a}}; x \in \text{set (tl a)} \rrbracket \Longrightarrow x \in \text{set a}$$

t.map [*simp*, *code*]:

$$\text{map fi } [] = []$$

$$\text{map fi (x21a \# x22)} = \text{fi x21a} \# \text{map fi x22}$$

t.disc_map_iff [*simp*]:

$$\overline{\text{null (map f a)}} = \overline{\text{null a}}$$

t.sel_map:

$$\neg \overline{\text{null a}} \Longrightarrow \text{hd (map f a)} = f (\text{hd a})$$

$$\neg \overline{\text{null a}} \Longrightarrow \text{tl (map f a)} = \text{map f (tl a)}$$

t.rel_inject [*simp*]:

$$\overline{\text{list_all2 R } [] []}$$

$$\overline{\text{list_all2 R (x21a \# x22) (y21 \# y22)}} = (\text{R x21a y21} \wedge \overline{\text{list_all2 R x22 y22}})$$

t.rel_distinct [*simp*]:

$$\neg \overline{\text{list_all2 R } [] (y21 \# y22)}$$

$$\neg \overline{\text{list_all2 R (y21 \# y22) []}}$$

t.rel_intros:

$$\overline{\text{list_all2 R } [] []}$$

$$\llbracket \text{R x21a y21}; \overline{\text{list_all2 R x22 y22}} \rrbracket \Longrightarrow \overline{\text{list_all2 R (x21a \# x22) (y21 \# y22)}}$$

t.rel_sel:

$$\overline{\text{list_all2 R a b}} = (\overline{\text{null a}} = \overline{\text{null b}} \wedge (\neg \overline{\text{null a}} \longrightarrow \neg \overline{\text{null b}} \longrightarrow \text{R (hd a) (hd b)} \wedge \overline{\text{list_all2 R (tl a) (tl b)}}))$$

In addition, equational versions of *t.rel_inject* and *rel_distinct* are registered with the [*code*] attribute.

The second subgroup consists of more abstract properties of the set functions, the map function, and the relator:

t.set_map:

$$\overline{\text{set (map f v)}} = f \text{ ' set v}$$

t.map_comp:

$$(\bigwedge z. z \in \text{set x} \Longrightarrow f z = g z) \Longrightarrow \text{map f x} = \text{map g x}$$

$t.map_cong$ [*fundef_cong*]:
 $\llbracket x = y; \bigwedge z. z \in set\ y \implies f\ z = g\ z \rrbracket \implies map\ f\ x = map\ g\ y$

$t.map_id$:
 $map\ id\ t = t$

$t.map_id0$:
 $map\ id = id$

$t.map_ident$:
 $map\ (\lambda x. x)\ t = t$

$t.rel_compp$:
 $\overline{list_all2}\ (R\ OO\ S) = list_all2\ R\ OO\ list_all2\ S$

$t.rel_conversep$:
 $\overline{list_all2}\ R^{--} = (list_all2\ R)^{--}$

$t.rel_eq$:
 $\overline{list_all2}\ op = =\ op =$

$t.rel_flip$:
 $\overline{list_all2}\ R^{--}\ a\ b = list_all2\ R\ b\ a$

$t.rel_mono$:
 $R \leq Ra \implies list_all2\ R \leq list_all2\ Ra$

2.4.3 Inductive Theorems

The inductive theorems are as follows:

$t.induct$ [*case_names* $C_1 \dots C_n$, *induct* t]:
 $\llbracket P\ []; \bigwedge x1\ x2. P\ x2 \implies P\ (x1\ \#\ x2) \rrbracket \implies P\ list$

$t_1 \dots t_m.induct$ [*case_names* $C_1 \dots C_n$]:
 Given $m > 1$ mutually recursive datatypes, this induction rule can be used to prove m properties simultaneously.

$t.rel_induct$ [*case_names* $C_1 \dots C_n$, *induct* $pred$]:
 $\llbracket list_all2\ R\ x\ y; Q\ []\ []; \bigwedge a21\ a22\ b21\ b22. \llbracket R\ a21\ b21; Q\ a22\ b22 \rrbracket \implies Q\ (a21\ \#\ a22)\ (b21\ \#\ b22) \rrbracket \implies Q\ x\ y$

$t_1 \dots t_m.rel_induct$ [*case_names* $C_1 \dots C_n$]:
 Given $m > 1$ mutually recursive datatypes, this induction rule can be used to prove m properties simultaneously.

$t.rec$ [*simp*, *code*]:
 $rec_list\ f1\ f2\ [] = f1$
 $rec_list\ f1\ f2\ (x21\ \#\ x22) = f2\ x21\ x22\ (rec_list\ f1\ f2\ x22)$

```

t.rec_o_map:
  rec_list h ha ∘ map g = rec_list h (λx xa. ha (g x) (map g xa))
t.size [simp, code]:
  size_list x [] = 0
  size_list x (x21 # x22) = x x21 + size_list x x22 + Nat.Suc 0
  size [] = 0
  size (x21 # x22) = size x22 + Nat.Suc 0
t.size_o_map:
  size_list f ∘ map g = size_list (f ∘ g)

```

For convenience, `datatype_new` also provides the following collection:

```

t.simps = t.inject t.distinct t.case t.rec t.map t.rel_inject
         t.rel_distinct t.set

```

2.5 Compatibility Issues

The command `datatype_new` has been designed to be highly compatible with the old `datatype`, to ease migration. There are nonetheless a few incompatibilities that may arise when porting to the new package:

- *The Standard ML interfaces are different.* Tools and extensions written to call the old ML interfaces will need to be adapted to the new interfaces. This concerns Quickcheck in particular. Whenever possible, it is recommended to use `datatype_compat` to register new-style datatypes as old-style datatypes.
- *The constants `t_case`, `t_rec`, and `t_size` are now called `case_t`, `rec_t`, and `size_t`.*
- *The recursor `rec_t` has a different signature for nested recursive datatypes.* In the old package, nested recursion through non-functions was internally reduced to mutual recursion. This reduction was visible in the type of the recursor, used by `primrec`. Recursion through functions was handled specially. In the new package, nested recursion (for functions and non-functions) is handled in a more modular fashion. The old-style recursor can be generated on demand using `primrec` if the recursion is via new-style datatypes, as explained in Section 3.1.5.
- *Accordingly, the induction rule is different for nested recursive datatypes.* Again, the old-style induction rule can be generated on demand using `primrec` if the recursion is via new-style datatypes, as explained in Section 3.1.5.

- *The internal constructions are completely different.* Proof texts that unfold the definition of constants introduced by **datatype** will be difficult to port.
- *Some theorems have different names.* For non-mutually recursive datatypes, the alias *t.inducts* for *t.induct* is no longer generated. For $m > 1$ mutually recursive datatypes, *t₁..._m.inducts(i)* has been renamed *t_i.induct* for each $i \in \{1, \dots, t\}$, and similarly the collection *t₁..._m.size* has been divided into *t₁.size*, ..., *t_m.size*.
- *The t.simps collection has been extended.* Previously available theorems are available at the same index.
- *Variables in generated properties have different names.* This is rarely an issue, except in proof texts that refer to variable names in the [*where ...*] attribute. The solution is to use the more robust [*of ...*] syntax.

In the other direction, there is currently no way to register old-style datatypes as new-style datatypes. If the goal is to define new-style datatypes with nested recursion through old-style datatypes, the old-style datatypes can be registered as a BNF (Section 6). If the goal is to derive discriminators and selectors, this can be achieved using **free_constructors** (Section 7).

3 Defining Recursive Functions

Recursive functions over datatypes can be specified using the **primrec** command, which supports primitive recursion, or using the more general **fun** and **function** commands. Here, the focus is on **primrec**; the other two commands are described in a separate tutorial [5].

3.1 Introductory Examples

Primitive recursion is illustrated through concrete examples based on the datatypes defined in Section 2.1. More examples can be found in the directory `~/src/HOL/BNF_Examples`.

3.1.1 Nonrecursive Types

Primitive recursion removes one layer of constructors on the left-hand side in each equation. For example:

```
primrec bool_of_trool :: "trool  $\Rightarrow$  bool" where
  "bool_of_trool Faalse  $\longleftrightarrow$  False" |
```

`“bool_of_troot True \longleftrightarrow True”`

primrec `the_list` :: `“'a option \Rightarrow 'a list”` **where**
`“the_list None = []”` |
`“the_list (Some a) = [a]”`

primrec `the_default` :: `“'a \Rightarrow 'a option \Rightarrow 'a”` **where**
`“the_default d None = d”` |
`“the_default _ (Some a) = a”`

primrec `mirrror` :: `“(‘a, ‘b, ‘c) triple \Rightarrow (‘c, ‘b, ‘a) triple”` **where**
`“mirrror (Triple a b c) = Triple c b a”`

The equations can be specified in any order, and it is acceptable to leave out some cases, which are then unspecified. Pattern matching on the left-hand side is restricted to a single datatype, which must correspond to the same argument in all equations.

3.1.2 Simple Recursion

For simple recursive types, recursive calls on a constructor argument are allowed on the right-hand side:

primrec `replicate` :: `“nat \Rightarrow 'a \Rightarrow 'a list”` **where**
`“replicate Zero _ = []”` |
`“replicate (Suc n) x = x # replicate n x”`

primrec `at` :: `“'a list \Rightarrow nat \Rightarrow 'a”` **where**
`“at (x # xs) j =`
`(case j of`
`Zero \Rightarrow x`
`| Suc j' \Rightarrow at xs j’)”`

primrec `tfold` :: `“(‘a \Rightarrow ‘b \Rightarrow ‘b) \Rightarrow (‘a, ‘b) tlist \Rightarrow ‘b”` **where**
`“tfold _ (TNil y) = y”` |
`“tfold f (TCons x xs) = f x (tfold f xs)”`

Pattern matching is only available for the argument on which the recursion takes place. Fortunately, it is easy to generate pattern-matching equations using the `simps_of_case` command provided by the theory `~/src/HOL/Library/Simps_Case_Conv`.

`simps_of_case at_simps: at_simps`

This generates the lemma collection `at_simps`:

$$\text{at } (x \# xs) \text{ Zero} = x \quad \text{at } (xa \# xs) (\text{nat.Suc } x) = \text{at } xs \ x$$

The next example is defined using **fun** to escape the syntactic restrictions imposed on primitively recursive functions. The **datatype_compat** command is needed to register new-style datatypes for use with **fun** and **function** (Section 2.2.2):

```
datatype_compat nat
fun at_least_two :: "nat  $\Rightarrow$  bool" where
  "at_least_two (Suc (Suc _))  $\longleftrightarrow$  True" |
  "at_least_two _  $\longleftrightarrow$  False"
```

3.1.3 Mutual Recursion

The syntax for mutually recursive functions over mutually recursive datatypes is straightforward:

```
primrec
  nat_of_even_nat :: "even_nat  $\Rightarrow$  nat" and
  nat_of_odd_nat :: "odd_nat  $\Rightarrow$  nat"
where
  "nat_of_even_nat Even_Zero = Zero" |
  "nat_of_even_nat (Even_Suc n) = Suc (nat_of_odd_nat n)" |
  "nat_of_odd_nat (Odd_Suc n) = Suc (nat_of_even_nat n)"

primrec
  eval_e :: "('a  $\Rightarrow$  int)  $\Rightarrow$  ('b  $\Rightarrow$  int)  $\Rightarrow$  ('a, 'b) exp  $\Rightarrow$  int" and
  eval_t :: "('a  $\Rightarrow$  int)  $\Rightarrow$  ('b  $\Rightarrow$  int)  $\Rightarrow$  ('a, 'b) trm  $\Rightarrow$  int" and
  eval_f :: "('a  $\Rightarrow$  int)  $\Rightarrow$  ('b  $\Rightarrow$  int)  $\Rightarrow$  ('a, 'b) fct  $\Rightarrow$  int"
where
  "eval_e  $\gamma$   $\xi$  (Term t) = eval_t  $\gamma$   $\xi$  t" |
  "eval_e  $\gamma$   $\xi$  (Sum t e) = eval_t  $\gamma$   $\xi$  t + eval_e  $\gamma$   $\xi$  e" |
  "eval_t  $\gamma$   $\xi$  (Factor f) = eval_f  $\gamma$   $\xi$  f" |
  "eval_t  $\gamma$   $\xi$  (Prod f t) = eval_f  $\gamma$   $\xi$  f + eval_t  $\gamma$   $\xi$  t" |
  "eval_f  $\gamma$  _ (Const a) =  $\gamma$  a" |
  "eval_f _  $\xi$  (Var b) =  $\xi$  b" |
  "eval_f  $\gamma$   $\xi$  (Expr e) = eval_e  $\gamma$   $\xi$  e"
```

Mutual recursion is possible within a single type, using **fun**:

```
fun
  even :: "nat  $\Rightarrow$  bool" and
  odd :: "nat  $\Rightarrow$  bool"
where
  "even Zero = True" |
  "even (Suc n) = odd n" |
  "odd Zero = False" |
  "odd (Suc n) = even n"
```

3.1.4 Nested Recursion

In a departure from the old datatype package, nested recursion is normally handled via the map functions of the nesting type constructors. For example, recursive calls are lifted to lists using *map*:

```
primrec at_ff :: "'a tree_ff ⇒ nat list ⇒ 'a" where
  "at_ff (Node_ff a ts) js =
    (case js of
     [] ⇒ a
    | j # js' ⇒ at (map (λt. at_ff t js') ts) j)"
```

The next example features recursion through the *option* type. Although *option* is not a new-style datatype, it is registered as a BNF with the map function *map_option*:

```
primrec sum_btree :: "('a::{zero,plus}) btree ⇒ 'a" where
  "sum_btree (BNode a lt rt) =
    a + the_default 0 (map_option sum_btree lt) +
    the_default 0 (map_option sum_btree rt)"
```

The same principle applies for arbitrary type constructors through which recursion is possible. Notably, the map function for the function type (\Rightarrow) is simply composition ($op \circ$):

```
primrec relabel_ft :: "('a ⇒ 'a) ⇒ 'a ftree ⇒ 'a ftree" where
  "relabel_ft f (FTLeaf x) = FTLeaf (f x)" |
  "relabel_ft f (FTNode g) = FTNode (relabel_ft f ∘ g)"
```

For convenience, recursion through functions can also be expressed using λ -abstractions and function application rather than through composition. For example:

```
primrec relabel_ft :: "('a ⇒ 'a) ⇒ 'a ftree ⇒ 'a ftree" where
  "relabel_ft f (FTLeaf x) = FTLeaf (f x)" |
  "relabel_ft f (FTNode g) = FTNode (λx. relabel_ft f (g x))"

primrec subtree_ft :: "'a ⇒ 'a ftree ⇒ 'a ftree" where
  "subtree_ft x (FTNode g) = g x"
```

For recursion through curried n -ary functions, n applications of $op \circ$ are necessary. The examples below illustrate the case where $n = 2$:

```
datatype_new 'a ftree2 = FTLeaf2 'a | FTNode2 "'a ⇒ 'a ⇒ 'a ftree2"

primrec relabel_ft2 :: "('a ⇒ 'a) ⇒ 'a ftree2 ⇒ 'a ftree2" where
  "relabel_ft2 f (FTLeaf2 x) = FTLeaf2 (f x)" |
  "relabel_ft2 f (FTNode2 g) = FTNode2 (op ∘ (op ∘ (relabel_ft2 f)) g)"

primrec relabel_ft2 :: "('a ⇒ 'a) ⇒ 'a ftree2 ⇒ 'a ftree2" where
```

```

“relabel_ft2 f (FTLeaf2 x) = FTLeaf2 (f x)” |
“relabel_ft2 f (FTNode2 g) = FTNode2 (λx y. relabel_ft2 f (g x y))”
primrec subtree_ft2 :: “'a ⇒ 'a ⇒ 'a ftree2 ⇒ 'a ftree2” where
  “subtree_ft2 x y (FTNode2 g) = g x y”

```

3.1.5 Nested-as-Mutual Recursion

For compatibility with the old package, but also because it is sometimes convenient in its own right, it is possible to treat nested recursive datatypes as mutually recursive ones if the recursion takes place through new-style datatypes. For example:

```

primrec
  at_ff :: “'a tree_ff ⇒ nat list ⇒ 'a” and
  ats_ff :: “'a tree_ff list ⇒ nat ⇒ nat list ⇒ 'a”
where
  “at_ff (Node_ff a ts) js =
    (case js of
     [] ⇒ a
     | j # js' ⇒ ats_ff ts j js’)” |
  “ats_ff (t # ts) j =
    (case j of
     Zero ⇒ at_ff t
     | Suc j' ⇒ ats_ff ts j’)”

```

Appropriate induction rules are generated as *at_ff.induct*, *ats_ff.induct*, and *at_ff_ats_ff.induct*. The induction rules and the underlying recursors are generated on a per-need basis and are kept in a cache to speed up subsequent definitions.

Here is a second example:

```


primrec
  sum_btree :: “(a::{zero,plus}) btree ⇒ 'a” and
  sum_btree_option :: “'a btree option ⇒ 'a”
where
  “sum_btree (BNode a lt rt) =
    a + sum_btree_option lt + sum_btree_option rt” |
  “sum_btree_option None = 0” |
  “sum_btree_option (Some t) = sum_btree t”

```


3.2 Command Syntax

3.2.1 primrec


primrec : *local_theory* → *local_theory*



pr-option



pr-equation



The **primrec** command introduces a set of mutually recursive functions over datatypes.

The syntactic entity *target* can be used to specify a local context, *fixes* denotes a list of names with optional type signatures, *thmdecl* denotes an optional name for the formula that follows, and *prop* denotes a HOL proposition [9].

The optional target is optionally followed by the following option:

- The *nonexhaustive* option indicates that the functions are not necessarily specified for all constructors. It can be used to suppress the warning that is normally emitted when some constructors are missing.

3.3 Recursive Default Values for Selectors

A datatype selector *un_D* can have a default value for each constructor on which it is not otherwise specified. Occasionally, it is useful to have the default value be defined recursively. This leads to a chicken-and-egg

situation, because the datatype is not introduced yet at the moment when the selectors are introduced. Of course, we can always define the selectors manually afterward, but we then have to state and prove all the characteristic theorems ourselves instead of letting the package do it.

Fortunately, there is a workaround that relies on overloading to relieve us from the tedium of manual derivations:

1. Introduce a fully unspecified constant $un_D_0 :: 'a$ using **consts**.
2. Define the datatype, specifying un_D_0 as the selector's default value.
3. Define the behavior of un_D_0 on values of the newly introduced datatype using the **overloading** command.
4. Derive the desired equation on un_D from the characteristic equations for un_D_0 .

The following example illustrates this procedure:

```

consts termi0 :: 'a

datatype_new ('a, 'b) tlist =
  TNil (termi: 'b)
| TCons (thd: 'a) (ttl: "('a, 'b) tlist")
where
  "ttl (TNil y) = TNil y"
| "termi (TCons _ xs) = termi0 xs"

overloading
  termi0 ≡ "termi0 :: ('a, 'b) tlist ⇒ 'b"
begin
primrec termi0 :: "('a, 'b) tlist ⇒ 'b" where
  "termi0 (TNil y) = y" |
  "termi0 (TCons x xs) = termi0 xs"
end

lemma termi_TCons[simp]: "termi (TCons x xs) = termi xs"
by (cases xs) auto

```

3.4 Compatibility Issues

The command **primrec**'s behavior on new-style datatypes has been designed to be highly compatible with that for old-style datatypes, to ease migration. There is nonetheless at least one incompatibility that may arise when porting to the new package:

- *Some theorems have different names.* For $m > 1$ mutually recursive functions, $f_1_ \dots _f_m.simps$ has been broken down into separate sub-collections $f_i.simps$.

4 Defining Codatatypes

Codatatypes can be specified using the **codatatype** command. The command is first illustrated through concrete examples featuring different flavors of corecursion. More examples can be found in the directory `~/src/HOL/BNF/Examples`. The *Archive of Formal Proofs* also includes some useful codatatypes, notably for lazy lists [6].

4.1 Introductory Examples

4.1.1 Simple Corecursion

Non-corecursive codatatypes coincide with the corresponding datatypes, so they are rarely used in practice. *Corecursive codatatypes* have the same syntax as recursive datatypes, except for the command name. For example, here is the definition of lazy lists:

```
codatatype (lset: 'a) llist =
  lnull: LNil
| LCons (lhd: 'a) (ltl: "'a llist")
for
  map: lmap
  rel: llist_all2
where
  "ltl LNil = LNil"
```

Lazy lists can be infinite, such as $LCons\ 0\ (LCons\ 0\ (\dots))$ and $LCons\ 0\ (LCons\ 1\ (LCons\ 2\ (\dots)))$. Here is a related type, that of infinite streams:

```
codatatype (sset: 'a) stream =
  SCons (shd: 'a) (stl: "'a stream")
for
  map: smap
  rel: stream_all2
```

Another interesting type that can be defined as a codatatype is that of the extended natural numbers:

```
codatatype enat = EZero | ESuc enat
```

This type has exactly one infinite element, $ESuc (ESuc (ESuc (\dots)))$, that represents ∞ . In addition, it has finite values of the form $ESuc (\dots (ESuc EZero) \dots)$.

Here is an example with many constructors:

```
codatatype 'a process =
  Fail
| Skip (cont: "'a process")
| Action (prefix: 'a) (cont: "'a process")
| Choice (left: "'a process") (right: "'a process")
```

Notice that the *cont* selector is associated with both *Skip* and *Action*.

4.1.2 Mutual Corecursion

The example below introduces a pair of *mutually corecursive* types:

```
codatatype even_enat = Even_EZero | Even_ESuc odd_enat
and odd_enat = Odd_ESuc even_enat
```

4.1.3 Nested Corecursion

The next examples feature *nested corecursion*:

```
codatatype 'a treeii = Nodeii (lblii: 'a) (subii: "'a treeii llist")
codatatype 'a treeis = Nodeis (lblis: 'a) (subis: "'a treeis fset")
codatatype 'a sm = SM (accept: bool) (trans: "'a  $\Rightarrow$  'a sm")
```

4.2 Command Syntax

4.2.1 codatatype

```
codatatype : local_theory  $\rightarrow$  local_theory
```


4.4.1 Coinductive Theorems

The coinductive theorems are listed below for 'a llist:

t.coinduct [consumes m , case_names $t_1 \dots t_m$,
case_conclusion $D_1 \dots D_n$, coinduct t]:

$$\llbracket R \text{ llist llist}'; \bigwedge \text{ llist llist}'. R \text{ llist llist}' \implies \text{lnull llist} = \text{lnull llist}' \wedge$$

$$(\neg \text{lnull llist} \longrightarrow \neg \text{lnull llist}' \longrightarrow \text{lhd llist} = \text{lhd llist}' \wedge R (\text{ltl llist}$$

$$(\text{ltl llist}')) \rrbracket \implies \text{llist} = \text{llist}'$$

t.strong_coinduct [consumes m , case_names $t_1 \dots t_m$,
case_conclusion $D_1 \dots D_n$]:

$$\llbracket R \text{ llist llist}'; \bigwedge \text{ llist llist}'. R \text{ llist llist}' \implies \text{lnull llist} = \text{lnull llist}' \wedge$$

$$(\neg \text{lnull llist} \longrightarrow \neg \text{lnull llist}' \longrightarrow \text{lhd llist} = \text{lhd llist}' \wedge (R (\text{ltl llist}$$

$$(\text{ltl llist}') \vee \text{ltl llist} = \text{ltl llist}')) \rrbracket \implies \text{llist} = \text{llist}'$$

t.rel_coinduct [consumes m , case_names $t_1 \dots t_m$,
case_conclusion $D_1 \dots D_n$, coinduct pred]:

$$\llbracket P \ x \ y; \bigwedge \text{ llist llist}'. P \ \text{llist llist}' \implies \text{lnull llist} = \text{lnull llist}' \wedge (\neg \text{lnull}$$

$$\text{llist} \longrightarrow \neg \text{lnull llist}' \longrightarrow R (\text{lhd llist}) (\text{lhd llist}') \wedge P (\text{ltl llist}) (\text{ltl}$$

$$\text{llist}') \rrbracket \implies \text{llist_all2 } R \ x \ y$$

$t_1 \dots t_m$.**coinduct** [case_names $t_1 \dots t_m$, case_conclusion $D_1 \dots D_n$]

$t_1 \dots t_m$.**strong_coinduct** [case_names $t_1 \dots t_m$,
case_conclusion $D_1 \dots D_n$]:

$t_1 \dots t_m$.**rel_coinduct** [case_names $t_1 \dots t_m$,
case_conclusion $D_1 \dots D_n$]:

Given $m > 1$ mutually corecursive codatatypes, these coinduction rules can be used to prove m properties simultaneously.

t.corec:

$p \ a \implies \text{corec_llist } p \ g21 \ q22 \ g221 \ g222 \ a = \text{LNil}$
 $\neg \ p \ a \implies \text{corec_llist } p \ g21 \ q22 \ g221 \ g222 \ a = \text{LCons } (g21 \ a) \ (\text{if}$
 $q22 \ a \ \text{then } g221 \ a \ \text{else } \text{corec_llist } p \ g21 \ q22 \ g221 \ g222 \ (g222 \ a))$

t.corec_code [code]:

$\text{corec_llist } p \ g21 \ q22 \ g221 \ g222 \ a = (\text{if } p \ a \ \text{then } \text{LNil} \ \text{else } \text{LCons}$
 $(g21 \ a) \ (\text{if } q22 \ a \ \text{then } g221 \ a \ \text{else } \text{corec_llist } p \ g21 \ q22 \ g221 \ g222$
 $(g222 \ a)))$

t.disc_corec:

$p \ a \implies \text{lnull } (\text{corec_llist } p \ g21 \ q22 \ g221 \ g222 \ a)$
 $\neg \ p \ a \implies \neg \ \text{lnull } (\text{corec_llist } p \ g21 \ q22 \ g221 \ g222 \ a)$

t.disc_corec_iff [simp]:

$\text{lnull } (\text{corec_llist } p \ g21 \ q22 \ g221 \ g222 \ a) = p \ a$
 $(\neg \ \text{lnull } (\text{corec_llist } p \ g21 \ q22 \ g221 \ g222 \ a)) = (\neg \ p \ a)$

```

t.sel_corec [simp]:
  ¬ p a ⇒ lhd (corec_llist p g21 q22 g221 g222 a) = g21 a
  ¬ p a ⇒ ltl (corec_llist p g21 q22 g221 g222 a) = (if q22 a then
    g221 a else corec_llist p g21 q22 g221 g222 (g222 a))

```

For convenience, **codatatype** also provides the following collection:

```

t.simps = t.inject t.distinct t.case t.disc_corec_iff t.sel_corec
          t.map t.rel_inject t.rel_distinct t.set

```

5 Defining Corecursive Functions

Corecursive functions can be specified using the **primcorec** and **primcorec-*ursive*** commands, which support primitive corecursion, or using the more general **partial_function** command. Here, the focus is on the first two. More examples can be found in the directory `~/src/HOL/BNF_Examples`.

Whereas recursive functions consume datatypes one constructor at a time, corecursive functions construct codatatypes one constructor at a time. Partly reflecting a lack of agreement among proponents of coalgebraic methods, Isabelle supports three competing syntaxes for specifying a function f :

- The *destructor view* specifies f by implications of the form

$$\dots \Longrightarrow is_C_j (f x_1 \dots x_n)$$

and equations of the form

$$un_C_j i (f x_1 \dots x_n) = \dots$$

This style is popular in the coalgebraic literature.

- The *constructor view* specifies f by equations of the form

$$\dots \Longrightarrow f x_1 \dots x_n = C_j \dots$$

This style is often more concise than the previous one.

- The *code view* specifies f by a single equation of the form

$$f x_1 \dots x_n = \dots$$

with restrictions on the format of the right-hand side. Lazy functional programming languages such as Haskell support a generalized version of this style.

All three styles are available as input syntax. Whichever syntax is chosen, characteristic theorems for all three styles are generated.

5.1 Introductory Examples

Primitive corecursion is illustrated through concrete examples based on the codatatypes defined in Section 4.1. More examples can be found in the directory `~/src/HOL/BNF_Examples`. The code view is favored in the examples below. Sections 5.1.5 and 5.1.6 present the same examples expressed using the constructor and destructor views.

5.1.1 Simple Corecursion

Following the code view, corecursive calls are allowed on the right-hand side as long as they occur under a constructor, which itself appears either directly to the right of the equal sign or in a conditional expression:

```
primcorec literate :: "('a ⇒ 'a) ⇒ 'a ⇒ 'a llist" where
  "literate g x = LCons x (literate g (g x))"
```

```
primcorec siterate :: "('a ⇒ 'a) ⇒ 'a ⇒ 'a stream" where
  "siterate g x = SCons x (siterate g (g x))"
```

The constructor ensures that progress is made—i.e., the function is *productive*. The above functions compute the infinite lazy list or stream $[x, g x, g (g x), \dots]$. Productivity guarantees that prefixes $[x, g x, g (g x), \dots, (g \wedge^k x)]$ of arbitrary finite length k can be computed by unfolding the code equation a finite number of times.

Corecursive functions construct codatatype values, but nothing prevents them from also consuming such values. The following function drops every second element in a stream:

```
primcorec every_snd :: "'a stream ⇒ 'a stream" where
  "every_snd s = SCons (shd s) (stl (stl s))"
```

Constructs such as *let-in*, *if-then-else*, and *case-of* may appear around constructors that guard corecursive calls:

```
primcorec lappend :: "'a llist ⇒ 'a llist ⇒ 'a llist" where
  "lappend xs ys =
    (case xs of
      LNil ⇒ ys
    | LCons x xs' ⇒ LCons x (lappend xs' ys))"
```

Pattern matching is not supported by **primcorec**. Fortunately, it is easy to generate pattern-matching equations using the **simps_of_case** command provided by the theory `~/src/HOL/Library/Simps_Case_Conv`.

```
simps_of_case lappend_simps: lappend.code
```

This generates the lemma collection *lappend_simps*:

$$\begin{aligned} \text{lappend } LNil \text{ } ys &= ys \\ \text{lappend } (LCons \text{ } xa \text{ } x) \text{ } ys &= LCons \text{ } xa \text{ } (\text{lappend } x \text{ } ys) \end{aligned}$$

Corecursion is useful to specify not only functions but also infinite objects:

```
primcorec infty :: enat where
  “infty = ESuc infty”
```

The example below constructs a pseudorandom process value. It takes a stream of actions (*s*), a pseudorandom function generator (*f*), and a pseudorandom seed (*n*):

```
primcorec
  random_process :: “’a stream ⇒ (int ⇒ int) ⇒ int ⇒ ’a process”
where
  “random_process s f n =
    (if n mod 4 = 0 then
      Fail
    else if n mod 4 = 1 then
      Skip (random_process s f (f n))
    else if n mod 4 = 2 then
      Action (shd s) (random_process (stl s) f (f n))
    else
      Choice (random_process (every_snd s) (f ∘ f) (f n))
      (random_process (every_snd (stl s)) (f ∘ f) (f (f n))))”
```

The main disadvantage of the code view is that the conditions are tested sequentially. This is visible in the generated theorems. The constructor and destructor views offer nonsequential alternatives.

5.1.2 Mutual Corecursion

The syntax for mutually corecursive functions over mutually corecursive data-types is unsurprising:

```
primcorec
  even_infty :: even_enat and
  odd_infty :: odd_enat
where
  “even_infty = Even_ESuc odd_infty” |
  “odd_infty = Odd_ESuc even_infty”
```


5.1.3 Nested Corecursion

The next pair of examples generalize the *literate* and *siterate* functions (Section 5.1.3) to possibly infinite trees in which subnodes are organized either as a lazy list ($tree_{ii}$) or as a finite set ($tree_{is}$). They rely on the map functions of the nesting type constructors to lift the corecursive calls:

primcorec $iterate_{ii} :: '(a \Rightarrow 'a\ llist) \Rightarrow 'a \Rightarrow 'a\ tree_{ii}$ **where**
 “ $iterate_{ii}\ g\ x = Node_{ii}\ x\ (lmap\ (iterate_{ii}\ g)\ (g\ x))$ ”

primcorec $iterate_{is} :: '(a \Rightarrow 'a\ fset) \Rightarrow 'a \Rightarrow 'a\ tree_{is}$ **where**
 “ $iterate_{is}\ g\ x = Node_{is}\ x\ (fimage\ (iterate_{is}\ g)\ (g\ x))$ ”

Both examples follow the usual format for constructor arguments associated with nested recursive occurrences of the datatype. Consider $iterate_{ii}$. The term $g\ x$ constructs an $'a\ llist$ value, which is turned into an $'a\ tree_{ii}\ llist$ value using $lmap$.

This format may sometimes feel artificial. The following function constructs a tree with a single, infinite branch from a stream:

primcorec $tree_{ii_of_stream} :: 'a\ stream \Rightarrow 'a\ tree_{ii}$ **where**
 “ $tree_{ii_of_stream}\ s =$
 $Node_{ii}\ (shd\ s)\ (lmap\ tree_{ii_of_stream}\ (LCons\ (stl\ s)\ LNil))$ ”

A more natural syntax, also supported by Isabelle, is to move corecursive calls under constructors:

primcorec $tree_{ii_of_stream} :: 'a\ stream \Rightarrow 'a\ tree_{ii}$ **where**
 “ $tree_{ii_of_stream}\ s =$
 $Node_{ii}\ (shd\ s)\ (LCons\ (tree_{ii_of_stream}\ (stl\ s))\ LNil)$ ”

The next example illustrates corecursion through functions, which is a bit special. Deterministic finite automata (DFAs) are traditionally defined as 5-tuples $(Q, \Sigma, \delta, q_0, F)$, where Q is a finite set of states, Σ is a finite alphabet, δ is a transition function, q_0 is an initial state, and F is a set of final states. The following function translates a DFA into a state machine:

primcorec $sm_of_dfa :: '(q \Rightarrow 'a \Rightarrow 'q) \Rightarrow 'q\ set \Rightarrow 'q \Rightarrow 'a\ sm$ **where**
 “ $sm_of_dfa\ \delta\ F\ q = SM\ (q \in F)\ (sm_of_dfa\ \delta\ F \circ \delta\ q)$ ”

The map function for the function type (\Rightarrow) is composition ($op \circ$). For convenience, corecursion through functions can also be expressed using λ -abstractions and function application rather than through composition. For example:

primcorec $sm_of_dfa :: '(q \Rightarrow 'a \Rightarrow 'q) \Rightarrow 'q\ set \Rightarrow 'q \Rightarrow 'a\ sm$ **where**
 “ $sm_of_dfa\ \delta\ F\ q = SM\ (q \in F)\ (\lambda a.\ sm_of_dfa\ \delta\ F\ (\delta\ q\ a))$ ”

primcorec $empty_sm :: 'a\ sm$ **where**

“*empty_sm* = *SM False* ($\lambda_.$ *empty_sm*)”

primcorec *not_sm* :: “*'a sm* \Rightarrow *'a sm*” **where**

“*not_sm M* = *SM* (\neg *accept M*) ($\lambda a.$ *not_sm* (*trans M a*))”

primcorec *or_sm* :: “*'a sm* \Rightarrow *'a sm* \Rightarrow *'a sm*” **where**

“*or_sm M N* =

SM (*accept M* \vee *accept N*) ($\lambda a.$ *or_sm* (*trans M a*) (*trans N a*))”

For recursion through curried n -ary functions, n applications of $op \circ$ are necessary. The examples below illustrate the case where $n = 2$:

codatatype (*'a, 'b*) *sm2* =

SM2 (*accept2: bool*) (*trans2: 'a* \Rightarrow *'b* \Rightarrow (*'a, 'b*) *sm2*)”

primcorec

sm2_of_dfa :: “(*'q* \Rightarrow *'a* \Rightarrow *'b* \Rightarrow *'q*) \Rightarrow *'q set* \Rightarrow *'q* \Rightarrow (*'a, 'b*) *sm2*”

where

“*sm2_of_dfa* δ *F q* = *SM2* ($q \in F$) ($op \circ (op \circ (sm2_of_dfa \delta F)) (\delta q)$)”

primcorec

sm2_of_dfa :: “(*'q* \Rightarrow *'a* \Rightarrow *'b* \Rightarrow *'q*) \Rightarrow *'q set* \Rightarrow *'q* \Rightarrow (*'a, 'b*) *sm2*”

where

“*sm2_of_dfa* δ *F q* = *SM2* ($q \in F$) ($\lambda a b.$ *sm2_of_dfa* δ *F* ($\delta q a b$))”

5.1.4 Nested-as-Mutual Corecursion

Just as it is possible to recurse over nested recursive datatypes as if they were mutually recursive (Section 3.1.5), it is possible to pretend that nested codatatypes are mutually corecursive. For example:

primcorec

iterate_{ii} :: “(*'a* \Rightarrow *'a llist*) \Rightarrow *'a* \Rightarrow *'a tree_{ii}*” **and**

iterates_{ii} :: “(*'a* \Rightarrow *'a llist*) \Rightarrow *'a llist* \Rightarrow *'a tree_{ii} llist*”

where

“*iterate_{ii}* *g x* = *Node_{ii}* *x* (*iterates_{ii}* *g* (*g x*))” |

“*iterates_{ii}* *g xs* =

(*case xs of*

LNil \Rightarrow *LNil*

| *LCons x xs'* \Rightarrow *LCons* (*iterate_{ii}* *g x*) (*iterates_{ii}* *g xs'*))”

Coinduction rules are generated as *iterate_{ii}.coinduct*, *iterates_{ii}.coinduct*, and *iterate_{ii}_iterates_{ii}.coinduct* and analogously for *strong_coinduct*. These rules and the underlying corecursors are generated on a per-need basis and are kept in a cache to speed up subsequent definitions.

5.1.5 Constructor View

The constructor view is similar to the code view, but there is one separate conditional equation per constructor rather than a single unconditional equation. Examples that rely on a single constructor, such as *literate* and *siterate*, are identical in both styles.

Here is an example where there is a difference:

```
primcorec lappend :: "'a llist  $\Rightarrow$  'a llist  $\Rightarrow$  'a llist" where
  "lnull xs  $\implies$  lnull ys  $\implies$  lappend xs ys = LNil" |
  "_  $\implies$  lappend xs ys = LCons (lhd (if lnull xs then ys else xs))
    (if xs = LNil then ltl ys else lappend (ltl xs) ys)"
```

With the constructor view, we must distinguish between the *LNil* and the *LCons* case. The condition for *LCons* is left implicit, as the negation of that for *LNil*.

For this example, the constructor view is slightly more involved than the code equation. Recall the code view version presented in Section 5.1.1. The constructor view requires us to analyze the second argument (*ys*). The code equation generated from the constructor view also suffers from this.

In contrast, the next example is arguably more naturally expressed in the constructor view:

```
primcorec
  random_process :: "'a stream  $\Rightarrow$  (int  $\Rightarrow$  int)  $\Rightarrow$  int  $\Rightarrow$  'a process"
where
  "n mod 4 = 0  $\implies$  random_process s f n = Fail" |
  "n mod 4 = 1  $\implies$ 
    random_process s f n = Skip (random_process s f (f n))" |
  "n mod 4 = 2  $\implies$ 
    random_process s f n = Action (shd s) (random_process (stl s) f (f n))" |
  "n mod 4 = 3  $\implies$ 
    random_process s f n = Choice (random_process (every_snd s) f (f n))
      (random_process (every_snd (stl s)) f (f n))"
```

Since there is no sequentiality, we can apply the equation for *Choice* without having first to discharge $n \bmod 4 \neq 0$, $n \bmod 4 \neq 1$, and $n \bmod 4 \neq 2$. The price to pay for this elegance is that we must discharge exclusivity proof obligations, one for each pair of conditions ($n \bmod 4 = i$, $n \bmod 4 = j$) with $i < j$. If we prefer not to discharge any obligations, we can enable the *sequential* option. This pushes the problem to the users of the generated properties.

5.1.6 Destructure View

The destructor view is in many respects dual to the constructor view. Conditions determine which constructor to choose, and these conditions are interpreted sequentially or not depending on the *sequential* option. Consider the following examples:

primcorec *literate* :: “ $(\prime a \Rightarrow \prime a) \Rightarrow \prime a \Rightarrow \prime a \text{ llist}$ ” **where**
 “ $\neg \text{lnull } (\text{literate } _ x)$ ” |
 “ $\text{lhd } (\text{literate } _ x) = x$ ” |
 “ $\text{ltl } (\text{literate } g x) = \text{literate } g (g x)$ ”

primcorec *siterate* :: “ $(\prime a \Rightarrow \prime a) \Rightarrow \prime a \Rightarrow \prime a \text{ stream}$ ” **where**
 “ $\text{shd } (\text{siterate } _ x) = x$ ” |
 “ $\text{stl } (\text{siterate } g x) = \text{siterate } g (g x)$ ”

primcorec *every_snd* :: “ $\prime a \text{ stream} \Rightarrow \prime a \text{ stream}$ ” **where**
 “ $\text{shd } (\text{every_snd } s) = \text{shd } s$ ” |
 “ $\text{stl } (\text{every_snd } s) = \text{stl } (\text{stl } s)$ ”

The first formula in the *local.literate* specification indicates which constructor to choose. For *local.siterate* and *local.every_snd*, no such formula is necessary, since the type has only one constructor. The last two formulas are equations specifying the value of the result for the relevant selectors. Corecursive calls appear directly to the right of the equal sign. Their arguments are unrestricted.

The next example shows how to specify functions that rely on more than one constructor:

primcorec *lappend* :: “ $\prime a \text{ llist} \Rightarrow \prime a \text{ llist} \Rightarrow \prime a \text{ llist}$ ” **where**
 “ $\text{lnull } xs \Longrightarrow \text{lnull } ys \Longrightarrow \text{lnull } (\text{lappend } xs ys)$ ” |
 “ $\text{lhd } (\text{lappend } xs ys) = \text{lhd } (\text{if } \text{lnull } xs \text{ then } ys \text{ else } xs)$ ” |
 “ $\text{ltl } (\text{lappend } xs ys) = (\text{if } xs = \text{LNil} \text{ then } \text{ltl } ys \text{ else } \text{lappend } (\text{ltl } xs) ys)$ ”

For a codatatype with n constructors, it is sufficient to specify $n - 1$ discriminator formulas. The command will then assume that the remaining constructor should be taken otherwise. This can be made explicit by adding

“ $_ \Longrightarrow \neg \text{lnull } (\text{lappend } xs ys)$ ”

to the specification. The generated selector theorems are conditional.

The next example illustrates how to cope with selectors defined for several constructors:

primcorec
random_process :: “ $\prime a \text{ stream} \Rightarrow (\text{int} \Rightarrow \text{int}) \Rightarrow \text{int} \Rightarrow \prime a \text{ process}$ ”
where
 “ $n \bmod 4 = 0 \Longrightarrow \text{random_process } s f n = \text{Fail}$ ” |

```

“n mod 4 = 1  $\implies$  is_Skip (random_process s f n)” |
“n mod 4 = 2  $\implies$  is_Action (random_process s f n)” |
“n mod 4 = 3  $\implies$  is_Choice (random_process s f n)” |
“cont (random_process s f n) = random_process s f (f n)” of Skip |
“prefix (random_process s f n) = shd s” |
“cont (random_process s f n) = random_process (stl s) f (f n)” of Action |
“left (random_process s f n) = random_process (every_snd s) f (f n)” |
“right (random_process s f n) = random_process (every_snd (stl s)) f (f n)”

```

Using the *of* keyword, different equations are specified for *cont* depending on which constructor is selected.

Here are more examples to conclude:

primcorec

```
even_infty :: even_enat and
```

```
odd_infty :: odd_enat
```

where

```
“even_infty  $\neq$  Even_EZero” |
```

```
“un_Even_ESuc even_infty = odd_infty” |
```

```
“un_Odd_ESuc odd_infty = even_infty”
```

primcorec iterate_{ii} :: “(‘a \implies ‘a llist) \implies ‘a \implies ‘a tree_{ii}” where

```
“lblii (iterateii g x) = x” |
```


```
“subii (iterateii g x) = lmap (iterateii g) (g x)”
```



5.2 Command Syntax

5.2.1 primcorec and primcorecursive

```
primcorec : local_theory  $\rightarrow$  local_theory
```

```
primcorecursive : local_theory  $\rightarrow$  proof(prove)
```



pcr-option*pcr-formula*

The **primcorec** and **primcorecursive** commands introduce a set of mutually corecursive functions over codatatypes.

The syntactic entity *target* can be used to specify a local context, *fixes* denotes a list of names with optional type signatures, *thmdecl* denotes an optional name for the formula that follows, and *prop* denotes a HOL proposition [9].

The optional target is optionally followed by one or both of the following options:

- The *sequential* option indicates that the conditions in specifications expressed using the constructor or destructor view are to be interpreted sequentially.
- The *exhaustive* option indicates that the conditions in specifications expressed using the constructor or destructor view cover all possible cases.

The **primcorec** command is an abbreviation for **primcorecursive** with *by auto?* to discharge any emerging proof obligations.

6 Introducing Bounded Natural Functors

The (co)datatype package can be set up to allow nested recursion through arbitrary type constructors, as long as they adhere to the BNF requirements and are registered as BNFs. It is also possible to declare a BNF abstractly without specifying its internal structure.

6.1 Bounded Natural Functors

Bounded natural functors (BNFs) are a semantic criterion for where (co)recursion may appear on the right-hand side of an equation [3, 8].

An n -ary BNF is a type constructor equipped with a map function (functorial action), n set functions (natural transformations), and an infinite cardinal bound that satisfy certain properties. For example, `'a llist` is a unary BNF. Its relator `llist_all2 :: ('a \Rightarrow 'b \Rightarrow bool) \Rightarrow 'a llist \Rightarrow 'b llist \Rightarrow bool` extends binary predicates over elements to binary predicates over parallel lazy lists. The cardinal bound limits the number of elements returned by the set function; it may not depend on the cardinality of `'a`.

The type constructors introduced by `datatype_new` and `codatatype` are automatically registered as BNFs. In addition, a number of old-style datatypes and non-free types are preregistered.

Given an n -ary BNF, the n type variables associated with set functions, and on which the map function acts, are *live*; any other variables are *dead*. Nested (co)recursion can only take place through live variables.

6.2 Introductory Examples

The example below shows how to register a type as a BNF using the `bnf` command. Some of the proof obligations are best viewed with the theory `Cardinal_Notations`, located in `~/src/HOL/Library`, imported.

The type is simply a copy of the function space `'d \Rightarrow 'a`, where `'a` is live and `'d` is dead. We introduce it together with its map function, set function, and relator.

```

typedef ('d, 'a) fn = "UNIV :: ('d  $\Rightarrow$  'a) set"
by simp

setup_lifting type_definition_fn

lift_definition map_fn :: "('a  $\Rightarrow$  'b)  $\Rightarrow$  ('d, 'a) fn  $\Rightarrow$  ('d, 'b) fn" is "op  $\circ$ " .
lift_definition set_fn :: "('d, 'a) fn  $\Rightarrow$  'a set" is range .

lift_definition
  rel_fn :: "('a  $\Rightarrow$  'b  $\Rightarrow$  bool)  $\Rightarrow$  ('d, 'a) fn  $\Rightarrow$  ('d, 'b) fn  $\Rightarrow$  bool"
is
  "rel_fun (op =)" .

bnf "('d, 'a) fn"
  map: map_fn
  sets: set_fn
  bd: "natLeq +c |UNIV :: 'd set|"

```

```

    rel: rel_fn
  proof -
    show "map_fn id = id"
      by transfer auto
  next
    fix F G show "map_fn (G ∘ F) = map_fn G ∘ map_fn F"
      by transfer (auto simp add: comp_def)
  next
    fix F f g
    assume "∧x. x ∈ set_fn F ⇒ f x = g x"
    thus "map_fn f F = map_fn g F"
      by transfer auto
  next
    fix f show "set_fn ∘ map_fn f = op ' f ∘ set_fn"
      by transfer (auto simp add: comp_def)
  next
    show "card_order (natLeq +c |UNIV :: 'd set| )"
      apply (rule card_order_csum)
      apply (rule natLeq_card_order)
      by (rule card_of_card_order_on)
  next
    show "cinfinitesimal (natLeq +c |UNIV :: 'd set| )"
      apply (rule cinfinitesimal_csum)
      apply (rule disjI1)
      by (rule natLeq_cinfinitesimal)
  next
    fix F :: "('d, 'a) fn"
    have "|set_fn F| ≤o |UNIV :: 'd set|" (is "_ ≤o ?U")
      by transfer (rule card_of_image)
    also have "?U ≤o natLeq +c ?U"
      by (rule ordLeq_csum2) (rule card_of_Card_order)
    finally show "|set_fn F| ≤o natLeq +c |UNIV :: 'd set|" .
  next
    fix R S
    show "rel_fn R OO rel_fn S ≤ rel_fn (R OO S)"
      by (rule, transfer) (auto simp add: rel_fun_def)
  next
    fix R
    show "rel_fn R =
      (BNF_Def.Grp {x. set_fn x ⊆ Collect (split R)} (map_fn fst))-- OO
      BNF_Def.Grp {x. set_fn x ⊆ Collect (split R)} (map_fn snd)"
    unfolding Grp_def fun_eq_iff relcompp.simps conversesep.simps
    apply transfer

```



```

unfolding rel_fun_def subset_iff image_iff
by auto (force, metis pair_collapse)
qed

print_theorems
print_bnfs

```

Using `print_theorems` and `print_bnfs`, we can contemplate and show the world what we have achieved.

This particular example does not need any nonemptiness witness, because the one generated by default is good enough, but in general this would be necessary. See `~/src/HOL/Basic_BNFs.thy`, `~/src/HOL/Library/FSet.thy`, and `~/src/HOL/Library/Multiset.thy` for further examples of BNF registration, some of which feature custom witnesses.

The next example declares a BNF axiomatically. This can be convenient for reasoning abstractly about an arbitrary BNF. The `bnf_axiomatization` command below introduces a type $(a, b, c) F$, three set constants, a map function, a relator, and a nonemptiness witness that depends only on a . (The type $a \Rightarrow (a, b, c) F$ of the witness can be read as an implication: If we have a witness for a , we can construct a witness for $(a, b, c) F$.) The BNF properties are postulated as axioms.

```

bnf_axiomatization (setA: 'a, setB: 'b, setC: 'c) F
  [wits: "'a  $\Rightarrow$  (a, b, c) F"]


print_theorems
print_bnfs

```

6.3 Command Syntax

6.3.1 bnf

`bnf` : *local_theory* \rightarrow *proof*(*prove*)




The **bnf** command registers an existing type as a bounded natural functor (BNF). The type must be equipped with an appropriate map function (functorial action). In addition, custom set functions, relators, and nonemptiness witnesses can be specified; otherwise, default versions are used.


The syntactic entity *target* can be used to specify a local context, *type* denotes a HOL type, and *term* denotes a HOL term [9].

6.3.2 bnf_axiomatization

bnf_axiomatization : *local_theory* \rightarrow *local_theory*



wit-types



The **bnf_axiomatization** command declares a new type and associated constants (map, set, relator, and cardinal bound) and asserts the BNF properties for these constants as axioms.


The syntactic entity *target* can be used to specify a local context, *name* denotes an identifier, *typefree* denotes fixed type variable (*'a*, *'b*, ...), and *mixfix* denotes the usual parenthesized mixfix notation [9].

Type arguments are live by default; they can be marked as dead by entering “*dead*” in front of the type variable (e.g., “(*dead 'a*)”) instead of an identifier for the corresponding set function. Witnesses can be specified by their types. Otherwise, the syntax of **bnf_axiomatization** is identical to the left-hand side of a **datatype_new** or **codatatype** definition.

The command is useful to reason abstractly about BNFs. The axioms are safe because there exist BNFs of arbitrary large arities. Applications must import the theory *BNF_Axiomatization*, located in the directory `~/src/HOL/Library`, to use this functionality.

6.3.3 print_bnfs

print_bnfs : *local_theory* →




7 Deriving Destructors and Theorems for Free Constructors

The derivation of convenience theorems for types equipped with free constructors, as performed internally by `datatype_new` and `codatatype`, is available as a stand-alone command called `free_constructors`.


7.1 Command Syntax

7.1.1 `free_constructors`

`free_constructors` : *local_theory* \rightarrow *proof(prove)*



fc-ctor



The `free_constructors` command generates destructor constants for freely constructed types as well as properties about constructors and destructors. It

also registers the constants and theorems in a data structure that is queried by various tools (e.g., **function**).

The syntactic entity *target* can be used to specify a local context, *name* denotes an identifier, *prop* denotes a HOL proposition, and *term* denotes a HOL term [9].

The syntax resembles that of **datatype_new** and **codatatype** definitions (Sections 2.2 and 4.2). A constructor is specified by an optional name for the discriminator, the constructor itself (as a term), and a list of optional names for the selectors.

Section 2.4 lists the generated theorems. For bootstrapping reasons, the generally useful [*fundef_cong*] attribute is not set on the generated *case_cong* theorem. It can be added manually using **declare**.

Acknowledgment

Tobias Nipkow and Makarius Wenzel encouraged us to implement the new (co)datatype package. Andreas Lochbihler provided lots of comments on earlier versions of the package, especially on the coinductive part. Brian Huffman suggested major simplifications to the internal constructions, many of which have yet to be implemented. Florian Haftmann and Christian Urban provided general advice on Isabelle and package writing. Stefan Milius and Lutz Schröder found an elegant proof that eliminated one of the BNF proof obligations. Andreas Lochbihler and Christian Sternagel suggested many textual improvements to this tutorial.

References

- [1] S. Berghofer and M. Wenzel. Inductive datatypes in HOL — lessons learned in Formal-Logic Engineering. In Y. Bertot, G. Dowek, A. Hirschowitz, C. Paulin, and L. Theys, editors, *Theorem Proving in Higher Order Logics: TPHOLs '99*, volume 1690 of *Lecture Notes in Computer Science*. Springer-Verlag, 1999.
- [2] J. C. Blanchette, J. Hölzl, A. Lochbihler, L. Panny, A. Popescu, and D. Traytel. Truly modular (co)datatypes for Isabelle/HOL. In G. Klein and R. Gamboa, editors, *ITP 2014*, volume 8558 of *LNCS*, pages 93–110. Springer, 2014.
- [3] J. C. Blanchette, A. Popescu, and D. Traytel. Witnessing (co)datatypes. <http://www21.in.tum.de/~blanchet/wit.pdf>, 2014.

- [4] M. J. C. Gordon, R. Milner, and C. P. Wadsworth. *Edinburgh LCF: A Mechanised Logic of Computation*. LNCS 78. Springer, 1979.
- [5] A. Krauss. *Defining Recursive Functions in Isabelle/HOL*. <http://isabelle.in.tum.de/doc/functions.pdf>.
- [6] A. Lochbihler. Coinductive. In G. Klein, T. Nipkow, and L. C. Paulson, editors, *The Archive of Formal Proofs*. <http://afp.sourceforge.net/entries/Coinductive.shtml>, Feb. 2010.
- [7] L. Panny, J. C. Blanchette, and D. Traytel. Primitively (co)recursive definitions for Isabelle/HOL. In *Isabelle Workshop 2014*, 2014.
- [8] D. Traytel, A. Popescu, and J. C. Blanchette. Foundational, compositional (co)datatypes for higher-order logic—Category theory applied to theorem proving. In *LICS 2012*, pages 596–605. IEEE, 2012.
- [9] M. Wenzel. *The Isabelle/Isar Reference Manual*. <http://isabelle.in.tum.de/doc/isar-ref.pdf>.